
MS SQL Report: 
Solarwinds Fix Disk Single Volume Usage with Trend 

 
System:   

 Windows 2008 R2,  
 MS SQL 2008 SP1,   
 Report Builder 2.0 

 
Data Source:  NetPerfMon database 
 

 
 
3 tables are used: 

1. VolumeUsage_Detail: This table keeps the volume usage for latest n1 days (n1 is 
Detailed Statistic Retention days set in Orion Polling Settings. Detailed statistics will be 
summarized into hourly statistics after configured n1 days. ) 

2. VolumeUsage_Hourly: This table keeps volume usage between n1 and n1+n2 days. (n2 
is Hourly Statistics Retention days set in Orion Polling Settings. Hourly statistics will be 
summarized into daily statistics after configured n2 days.) 

3. VolumeUsage_Daily: This table keeps volume usage before n1+n2 days and up to n3 
days. (n3 is Daily Statistics Retention days set in Orion Polling Settings. Daily statistics 
will be deleted from the database after configured n3 days.) One volume one record per 
day. 

 

 



 



 



 

SELECT Top 1 [DateTime]FROM [NetPerfMon].[dbo].[VolumeUsage_Daily] order by [DateTime] asc 

 

 

SELECT Top 1 [DateTime]FROM [NetPerfMon].[dbo].[VolumeUsage_Daily] order by [DateTime] Desc 



 

SELECT Top 1 [DateTime] FROM [NetPerfMon].[dbo].[VolumeUsage_Hourly] order by [DateTime] Asc 

 

SELECT Top 1 [DateTime] FROM [NetPerfMon].[dbo].[VolumeUsage_Hourly] order by [DateTime] Desc 



 

SELECT Top 1 [DateTime] FROM [NetPerfMon].[dbo].[VolumeUsage_Detail] order by [DateTime] Asc 

 

SELECT Top 1 [DateTime] FROM [NetPerfMon].[dbo].[VolumeUsage_Detail] order by [DateTime] Desc 



 

SELECT  
 Round((Volumes.VolumeSize/1024/1024/1024),2) AS VolumeSize, 
 Round((Volumes.VolumeSpaceUsed/1024/1024/1024),2) AS VolumeSpaceUsed, 
 Round((Volumes.VolumeSpaceAvailable/1024/1024/1024),2) AS VolumeSpaceAvailable, 
 Round(Volumes.VolumePercentUsed,1) AS VolumePercentUsed 
FROM  
 (Volumes INNER JOIN Nodes ON Volumes.NodeID = Nodes.NodeID) 
WHERE  
 (Nodes.Caption = @NodeName) AND (Volumes.Caption = @VolumeName) 
 
 
 
 
  



 
 
SELECT  
 DISTINCT Nodes.NodeID AS NodeID, 
 Nodes.Caption AS NodeName 
FROM  
 (Nodes INNER JOIN Volumes ON Nodes.NodeID = Volumes.NodeID) 
WHERE  
 Volumes.VolumeType ='Fixed Disk' 
Order by Nodes.Caption 



 
 
SELECT  
 DISTINCT  Volumes.VolumeID AS VolumeID, 
 Volumes.Caption AS DiskCaption 
FROM  
 (Volumes INNER JOIN Nodes ON Volumes.NodeID = Nodes.NodeID) 
WHERE  
 (Volumes.VolumeType ='Fixed Disk') AND (Nodes.Caption = @NodeName) 
 
Order by VolumeID 
 



 
 
SELECT DISTINCT VolumeUsage_Daily.DateTime AS DateTime 
FROM VolumeUsage_Daily  
 INNER JOIN Nodes ON VolumeUsage_Daily.NodeID = Nodes.NodeID 
WHERE Nodes.Caption = @NodeName 
UNION 
SELECT ( 
  Case When  
    (SELECT COUNT(VolumeUsage_Daily.DateTime) 
     FROM VolumeUsage_Daily  
     INNER JOIN Nodes ON VolumeUsage_Daily.NodeID = Nodes.NodeID 
     WHERE Nodes.Caption = @NodeName) < 1  
   THEN 
     @DailyEndTime 
   END 
)AS DateTime  
ORDER BY DateTime 
 



 
 
The query below collects and calculates a single volume usage between chosen start date and now, one 
record per day.  Then it uses linear regression method to generate trend usage report.  If strong linear 
relationship is found, the report will give the predicted date on which the volume will be full (if the date 
is not more than 365 days) and further 60 days data will be shown on the chart. 
 
The correlation coefficient always takes a value between -1 and 1 

1. Value 0 indicates no linear relationship 
2. Values 1 or -1 +1 indicates a perfect positive/negative linear relationship 
3. Values between 0 and 0.3 (0 and -0.3) indicate a weak positive (negative) linear relationship 
4. Values between 0.3 and 0.7 (0.3 and -0.7) indicate a moderate positive (negative) linear 

relationship 
5. Values between 0.7 and 1.0 (-0.7 and -1.0) indicate a strong positive (negative) linear 

relationship  
 
 
 
Query:   
 
DECLARE @LRLS_AvgX float(53), 
        @LRLS_AvgY float(53), 
        @LRLS_RowCountTotal int, 
        @LRLS_aValue float(53), 
        @LRLS_bValue float(53), 
        @LRLS_rValue float(53), 
        @theLastDateTime datetime, 
        @theLastDateTimeX int, 
        @LRLS_PredictFullDayX int, 
        @iCounter int 
; 
 
-- Temp Table 1: ##TempTbVolumeUsage 
 



SELECT TempTb01.DiskSize_GB, TempTb01.DateTime, TempTb01.DateTimeX,  
       TempTb01.RowCountN, TempTb01.DiskUsed_GB, TempTb01.PercentDiskUsed  
       INTO ##TempTbVolumeUsage 
FROM 
( 
(SELECT  
 Round(VolumeUsage_Daily.DiskSize/1024/1024/1024,3) AS DiskSize_GB, 
 VolumeUsage_Daily.DateTime AS DateTime, 
 DATEDIFF(day, @ReportStartTime, VolumeUsage_Daily.DateTime) AS DateTimeX, 
 1 AS RowCountN, 
 Round(VolumeUsage_Daily.AvgDiskUsed/1024/1024/1024, 3) AS DiskUsed_GB, 
 Round(VolumeUsage_Daily.PercentDiskUsed,2) AS PercentDiskUsed 
FROM  
 (Nodes INNER JOIN Volumes ON (Nodes.NodeID = Volumes.NodeID))  INNER JOIN 
VolumeUsage_Daily ON (Volumes.VolumeID = VolumeUsage_Daily.VolumeID) 
WHERE  
 (VolumeUsage_Daily.DateTime BETWEEN @ReportStartTime AND @DailyEndTime) AND 
 (Nodes.Caption = @NodeName AND Volumes.VolumeType ='Fixed Disk' AND Volumes.Caption = 
@VolumeName) 
) 
UNION 
(SELECT  
  ROUND(AVG(VolumeUsage_Hourly.DiskSize)/1024/1024/1024,3) AS DiskSize_GB, 
  CONVERT(date,VolumeUsage_Hourly.DateTime) AS DateTime, 
  DATEDIFF(day,@ReportStartTime, CONVERT(date,VolumeUsage_Hourly.DateTime)) AS DateTimeX, 
  1 AS RowCountN, 
  ROUND(AVG(VolumeUsage_Hourly.AvgDiskUsed)/1024/1024/1024,3) AS DiskUsed_GB, 
  ROUND(AVG(VolumeUsage_Hourly.PercentDiskUsed),1) AS PercentDiskUsed 
FROM  
 (Nodes INNER JOIN Volumes ON (Nodes.NodeID = Volumes.NodeID))  INNER JOIN 
VolumeUsage_Hourly ON (Volumes.VolumeID = VolumeUsage_Hourly.VolumeID) 
WHERE 
   (Nodes.Caption = @NodeName AND Volumes.VolumeType ='Fixed Disk' AND Volumes.Caption = 
@VolumeName) AND  
   (CONVERT(date,VolumeUsage_Hourly.DateTime) BETWEEN  DATEADD(dd, DATEDIFF(dd, 0, 
@HourlyStartTime), 0) AND DATEADD(dd, DATEDIFF(dd, 0, @HourlyEndTime), 0)  ) AND 
   (CONVERT(time(0),VolumeUsage_Hourly.DateTime) BETWEEN CONVERT(time(0),'00:00:00') AND 
CONVERT(time(0),'02:00:00') )  
GROUP BY CONVERT(date,VolumeUsage_Hourly.DateTime), Nodes.NodeID, Volumes.VolumeID, 
Nodes.Caption, Volumes.Caption 
) 
UNION 
(SELECT  
  ROUND(AVG(VolumeUsage_Detail.DiskSize)/1024/1024/1024,3) AS DiskSize_GB, 
  CONVERT(date,VolumeUsage_Detail.DateTime) AS DateTime, 
  DATEDIFF(day, @ReportStartTime, CONVERT(date,VolumeUsage_Detail.DateTime)) AS DateTimeX, 
  1 AS RowCountN, 
  ROUND(AVG(VolumeUsage_Detail.AvgDiskUsed)/1024/1024/1024,3) AS DiskUsed_GB, 
  ROUND(AVG(VolumeUsage_Detail.PercentDiskUsed),1) AS PercentDiskUsed 
FROM  
 (Nodes INNER JOIN Volumes ON (Nodes.NodeID = Volumes.NodeID))  INNER JOIN 
VolumeUsage_Detail ON (Volumes.VolumeID = VolumeUsage_Detail.VolumeID) 
WHERE 
   (Nodes.Caption = @NodeName AND Volumes.VolumeType ='Fixed Disk'  AND Volumes.Caption = 
@VolumeName) AND  
   (CONVERT(date,VolumeUsage_Detail.DateTime) BETWEEN  DATEADD(dd, DATEDIFF(dd, 0, 
@DetailStartTime), 0) AND DATEADD(dd, DATEDIFF(dd, 0, @DetailEndTime), 0))  AND 
   (CONVERT(time(0),VolumeUsage_Detail.DateTime) BETWEEN CONVERT(time(0),'00:00:00') AND 
CONVERT(time(0),'00:15:00') )  
GROUP BY CONVERT(date,VolumeUsage_Detail.DateTime), Nodes.NodeID, Volumes.VolumeID, 
Nodes.Caption, Volumes.Caption 



) 
) TempTb01; 
 
SET @iCounter = 1; 
SET @theLastDateTime = (SELECT TOP 1 DateTime FROM ##TempTbVolumeUsage Order BY 
DateTime DESC); 
SET @theLastDateTimeX = (SELECT TOP 1 DateTimeX FROM ##TempTbVolumeUsage Order BY 
DateTimeX DESC); 
SET @LRLS_AvgX = (SELECT AVG(DateTimeX+0.000) FROM ##TempTbVolumeUsage); 
SET @LRLS_AvgY = (SELECT AVG(DiskUsed_GB) FROM ##TempTbVolumeUsage); 
SET @LRLS_RowCountTotal = (SELECT SUM(RowCountN+0.0) FROM ##TempTbVolumeUsage); 
SET @LRLS_aValue = ( 
              SELECT  
               CASE COALESCE(SUM((##TempTbVolumeUsage.DateTimeX - 
@LRLS_AvgX)*(##TempTbVolumeUsage.DateTimeX - @LRLS_AvgX)),0) 
               WHEN 0 THEN 0 
               ELSE ROUND(SUM((##TempTbVolumeUsage.DateTimeX - 
@LRLS_AvgX)*(##TempTbVolumeUsage.DiskUsed_GB - @LRLS_AvgY)) / 
SUM((##TempTbVolumeUsage.DateTimeX - @LRLS_AvgX)*(##TempTbVolumeUsage.DateTimeX - 
@LRLS_AvgX)),4) 
               END 
              FROM ##TempTbVolumeUsage 
               ); 
SET @LRLS_bValue = @LRLS_AvgY - @LRLS_aValue * @LRLS_AvgX; 
SET @LRLS_rValue = ( 
               SELECT  
                CASE COALESCE((SUM((##TempTbVolumeUsage.DateTimeX -
@LRLS_AvgX)*(##TempTbVolumeUsage.DateTimeX -
@LRLS_AvgX))*SUM((##TempTbVolumeUsage.DiskUsed_GB-
@LRLS_AvgY)*(##TempTbVolumeUsage.DiskUsed_GB-@LRLS_AvgY))),0) 
  WHEN  0 THEN 0 
  ELSE  
               ROUND((SUM((##TempTbVolumeUsage.DateTimeX -
@LRLS_AvgX)*(##TempTbVolumeUsage.DiskUsed_GB-
@LRLS_AvgY)))/SQRT(SUM((##TempTbVolumeUsage.DateTimeX -
@LRLS_AvgX)*(##TempTbVolumeUsage.DateTimeX -
@LRLS_AvgX))*SUM((##TempTbVolumeUsage.DiskUsed_GB-
@LRLS_AvgY)*(##TempTbVolumeUsage.DiskUsed_GB-@LRLS_AvgY))),4)  
  END 
               FROM ##TempTbVolumeUsage 
                ); 
SET @LRLS_PredictFullDayX = ( 
               SELECT 
                 CASE WHEN (@LRLS_aValue > 0.0 AND @LRLS_rValue > 0.70) 
                      THEN CAST(((@VolumeSize-@LRLS_bValue)/@LRLS_aValue - @theLastDateTimeX) AS 
int) 
                      ELSE 0 
                 END 
               ); 
 
-- TEST: SELECT @LRLS_AvgX as avgX, @LRLS_AvgY as avgY, @LRLS_RowCountN as countX, 
@LRLS_aValue as aValue,  @LRLS_bValue as bValue, @LRLS_rValue as rVlaue 
 
-- Temp Table 2: predict data for another 60 days 
 
CREATE TABLE #TempTbVolumeUsageFuture( 
             DiskSize_GB float(53), DateTime datetime, DateTimeX int, RowCountN int, DiskUsed_GB 
float(53),  
             PercentDiskUsed float(53), PredictUsedSpace float(53), PredictUsedPercent float(53), 
RowCountTotal int, LRLS_aValue float(53), LRLS_rValue float(53), LRLS_bValue float(53), 
LastDateTimeX int, LRLS_PredictFullDayX int) 



WHILE (@iCounter < 61) 
BEGIN 
   INSERT INTO #TempTbVolumeUsageFuture VALUES( 
 @VolumeSize, 
 DATEADD(dd, @iCounter, @theLastDateTime), 
        @theLastDateTimeX + @iCounter, 
 1, 
        NULL, 
 NULL, 
 ROUND(@LRLS_aValue * (@theLastDateTimeX + @iCounter) + @LRLS_bValue,3), 
 CASE WHEN @VolumeSize = 0 THEN 0.0 ELSE ROUND(((@LRLS_aValue * 
(@theLastDateTimeX + @iCounter) + @LRLS_bValue)/@VolumeSize)*100.00,1) END,  
        @LRLS_RowCountTotal, 
 @LRLS_aValue, 
 @LRLS_rValue, 
        @LRLS_bValue, 
        @theLastDateTimeX, 
        @LRLS_PredictFullDayX 
   ) 
   SET @iCounter = @iCounter + 1; 
END 
 
-- TEST: SELECT * FROM #TempTbVolumeUsageFuture 
 
( 
SELECT *,  
       Round((DateTimeX*@LRLS_aValue+@LRLS_bValue),3) AS PredictUsedSpace, 
       CASE WHEN DiskSize_GB = 0 THEN 0.0 ELSE 
Round(((DateTimeX*@LRLS_aValue+@LRLS_bValue)/DiskSize_GB)*100.00,1) END AS 
PredictUsedPercent, 
       @LRLS_RowCountTotal AS RowCountTotal, 
       @LRLS_aValue AS LRLS_aValue, 
       @LRLS_rValue AS LRLS_rValue, 
       @LRLS_bValue AS LRLS_bValue, 
       @theLastDateTimeX AS LastDateTimeX, 
       @LRLS_PredictFullDayX AS LRLS_PredictFullDayX   
FROM ##TempTbVolumeUsage 
) 
UNION ALL 
( 
 SELECT * FROM #TempTbVolumeUsageFuture 
) 
 
; 
DROP TABLE ##TempTbVolumeUsage, #TempTbVolumeUsageFuture 
 
 
The Charts in the Report: 

 



 



 
 
Expressions used in reports: 
 

 
 
=DateDiff("d", Parameters!ReportStartTime.Value,Parameters!DetailEndTime.Value) 
 
 

 
 
 



E1: 
=IIF(Abs(First(Fields!LRLS_rValue.Value, "TbVolumeUsageTrend"))>0.70, "Strong Linear Trend!", 
IIF(Abs(First(Fields!LRLS_rValue.Value, "TbVolumeUsageTrend"))>0.30, "Moderate Linear Trend.", 
"Weak Linear Trend.")) 
 
E2: 
=IIF((First(Fields!LRLS_PredictFullDayX.Value, "TbVolumeUsageTrend"))=0, "", 
IIF((First(Fields!LRLS_PredictFullDayX.Value, "TbVolumeUsageTrend"))<365, "", "Plenty of disk space!")) 
 
E3: 
=IIF((First(Fields!LRLS_PredictFullDayX.Value, "TbVolumeUsageTrend"))=0, "", 
IIF((First(Fields!LRLS_PredictFullDayX.Value, "TbVolumeUsageTrend"))<365, 
First(Fields!LRLS_PredictFullDayX.Value, "TbVolumeUsageTrend"), "")) 
 
E4: 
=IIF((First(Fields!LRLS_PredictFullDayX.Value, "TbVolumeUsageTrend"))>0 AND 
(First(Fields!LRLS_PredictFullDayX.Value, "TbVolumeUsageTrend"))<365, "  Days later ( ", "") 
 
E5: 
=IIF((First(Fields!LRLS_PredictFullDayX.Value, "TbVolumeUsageTrend"))>0 AND 
(First(Fields!LRLS_PredictFullDayX.Value, "TbVolumeUsageTrend"))<365, 
DateAdd("d",First(Fields!LRLS_PredictFullDayX.Value, 
"TbVolumeUsageTrend"),Parameters!DetailEndTime.Value), "") 
 
E6: 
=IIF((First(Fields!LRLS_PredictFullDayX.Value, "TbVolumeUsageTrend"))>0 AND 
(First(Fields!LRLS_PredictFullDayX.Value, "TbVolumeUsageTrend"))<365, "  ) volume  will be FULL!", "") 
 
E7: 
=First(Fields!LRLS_aValue.Value, "TbVolumeUsageTrend") 
 
E8: 
=First(Fields!LRLS_rValue.Value, "TbVolumeUsageTrend") 
 
 
Displayed Table: 
 

 

 


